INDENG 240 : Homework 1

Arnaud Minondo

September 19, 2022

Production Planning Problem

My model file : pb1.mod

```
param n;
param order{i in 1..n};

param costPerTon{i in 1..n};
param inventoryCondition{i in 1..n};


param capacity ;
param invCostPerTon ;
param startInventory ;
param prodChangeCost ;


param matrix{i in 1..n,j in 1..n};


var production{i in 1..n}>=0;
var u{j in 1..(n-1)} >= 0;



minimize cost: (sum{i in 1..n} production[i]*costPerTon[i])
+ prodChangeCost*(sum{j in 1..(n-1)}u[j])
+ invCostPerTon*(3*(production[1]-order[1]+startInventory)+
2*(production[2]-order[2])+(production[3]- order[3]));

s.t. realLife{i in 1..n}: startInventory +
 (sum{j in 1..n}(production[j]-order[j])*matrix[i,j]) >= inventoryCondition[i];
s.t. plantCapacity{i in 1..n} : production[i]<=capacity;
s.t. absolute1{i in 1..(n-1)}: u[i]>=production[i]-production[i+1];
s.t. absolute2{i in 1..(n-1)}: u[i]>=production[i+1]-production[i];
```

My data file : pb1.dat

```
param n := 4;

param order :=
1 2400
2 2000
3 2700
4 2500
;

param costPerTon :=
1 7400
2 7500
3 7600
4 7800
;

param inventoryCondition :=
1 0
2 0
3 0
4 1500
;
param matrix:
1 2 3 4 :=
1 1 0 0 0
2 1 1 0 0
3 1 1 1 0
4 1 1 1 1
;
param capacity := 4000;
param invCostPerTon := 120;
param startInventory := 1000;
param prodChangeCost := 50;
```

My run file : pb1.run

```
reset;

model pb1.mod;

data pb1.dat;

option solver cplex;

solve;

display production, cost;

display production, cost > answer_pb1.txt;
```

The answer for the problem is : production = (2525,2525,2525,2525) with a cost of 77017500.

Betting Problem

My model file : pb2.mod

```
set SCENS;
set VARS;




param matrix {SCENS, VARS};

param amount_bet;

var strategy{VARS} >= 0;
var w >= 0;

maximize win: w;


s.t. budget: sum{i in VARS}strategy[i] <= amount_bet;
s.t. scen{j in SCENS}: w <= sum{i in VARS}matrix[j, i]*strategy[i];
```

My data file : pb2.dat

```
set SCENS = AA ABA ABB BB BAB BAA ;
set VARS = x1a x2a x3a x1b x2b x3b ya yb;

param amount_bet := 100;


param matrix:
    x1a   x1b   x2a  x2b   x3a   x3b   ya    yb :=
AA  0.67   -1   0.67 -1     0     0    0.4   -1
ABA 0.67   -1   -1   1.5   0.67  -1    0.4   -1
ABB 0.67   -1   -1   1.5   -1    1.5   -1    2.5
BB  -1    1.5   -1   1.5    0     0    -1    2.5
BAB -1    1.5   0.67 -1    -1    1.5   -1    2.5
BAA -1    1.5   0.67 -1    0.67  -1    0.4   -1
;
```

My run file : pb2.run

```
reset;

model pb2.mod;

data pb2.dat;

option solver cplex;
solve;

display strategy, win;

display strategy, win > pb2_answer.txt;
```

The optimal solution of this problem is strategy = x1a 17.6 x1b 0 x2a 17.6 x2b 0 x3a 43.8685 x3b 0 ya 0 yb 20.9316 where xia represents the amount bet on 'a' wins the i-th set, xib represent the amount bet on 'b' wins the i-th set and ya is the amount bet on 'a' wins the match and yb is the amount bet on 'b' wins the match. You are ensured to win at least $2.65.

Regression Problem

My model file : pb3.mod

```
param n;
param grades{i in 1..n};

param degree;

param hoursWorked{i in 1..n, j in 1..degree};


var u{i in 1..n};
var coefficients{i in 1..degree};

minimize Error :sum{i in 1..n}u[i];

s.t. absolute1{i in 1..n}:
u[i] >= (sum{j in 1..degree}(coefficients[j]*hoursWorked[i,j]))-grades[i];
s.t. absolute2{i in 1..n}:
u[i] >= (-1)*(sum{j in 1..degree}(coefficients[j]*hoursWorked[i,j])-grades[i]);
```

My data file : pb3.dat

```
param n := 8;

#bias introduced in the input
param hoursWorked :1 2 :=
1 4 1
2 9 1
3 10 1
4 14 1
5 4 1
6 7 1
7 12 1
8 3 1;

param grades :=
1 75
2 82
3 92
4 100
5 68
6 88
7 95
8 77;

param degree := 2;
```

My run file : pb3.run

```
reset;

model pb3.mod;

data pb3.dat;

option solver cplex;
solve;

display coefficients,Error;
display coefficients,Error>answer_pb3.txt;
```

The answer for this problem is coefficients=(2.5,65) with an Error of 24.5. The prediction for a student having worked only two hours is 2.5*2+65 = 70.

Solving LP Graphically

a.

I draw the conditions on the graph below. After the theorem of linear optimization, the optimal solution is on a border. As the function to maximize is $f(x, y) = 3x + y$ it is most likely to be optimal for higher x. benefit at the first critical point : $\begin{cases} y = 0 \\ 2x + y = 6 \end{cases}$ so $\begin{cases} x \quad = 3 \\ \quad y = 0 \end{cases}$ with $f(x, y) = 9$.
We need to check if the neighbours of this point are better or worse than this point :
$\begin{cases} 2x + y = 6 \\ x + y = 4 \end{cases}$ so $\begin{cases} x \quad = 2 \\ \quad y = 2 \end{cases}$, whith a cost function $f(2, 2) = 8$.
And obviously, (0,0) is not the maximizing point so (3,0) is the optimal solution.



b.

Every borders are the same for this variant of the problem, only the cost function changes : $f(x,y) = x + y$
As $x + y \leq 4$ then $max_{x,y}(f(x,y)) <= 4$ and x=2,y=2 verify this so $x = 2, y = 2$ is one optimal solution but there exist plenty of them, for example : $x = 0.5, y = 3.5$.



c.

The borders condition have changed but there is at the same time : $\begin{cases} -x + \quad y \leq -1 \\ -x + y \geq 1 \end{cases}$ which is impossible

d.

For this problem you can see in the plot, it is unbounded. Let's suppose there exist an optimal solution (x_0, y_0) such that $\forall (x, y) \in \mathbb{R}^2, f(x, y) \leq f(x_0, y_0) = x_0 + y_0$. Define $x_1 = x_0 + 1, y_1 = y_0 + 1, x_1 - y_1 = x_0 - y_0$ so (x_1, y_1) is a valid couple as it verifies $-1 \leq x_1 - y_1 \leq 1$. Moreover : $f(x_1, y_1) = x_0 + y_0 + 2 = f(x_0, y_0) + 2 > f(x_0, y_0)$ which contradicts the fact that (x_0, y_0) is the optimal couple. So there can't exist a maximum for this function under these limit conditions.