Case1 Calgary Desk Company Report

October 2017

Weixi (Sisi) Chen
Yuan (Jessica) Gu
Lin (Aaron) Lu
Xinning Yu

Introduction

This report is to recommend a production schedule to Calgary Desk Company for September. This is a real production problem that is important to the company since the amount of production will determine if CALDESCO will have financial gain or loss in the future. The aim is trying to maximize the profit under multiple constraints, such as labor capacity, material requirements, and company policy/quotas. Based on the data given, our team decided to maximize the total profit, which is the multiplication of individual profit of each type of table and the number of each type of table that should be produced, under the following constraints:

1. Customer Orders

   The amount of desks produced in September should be greater or equal to the amount that ordered by CALDESCO's clients.

2. Production Line Availability

   There are three production lines in CALDESCO and each one has its own time limitation: 9,600 minutes for Line 1, 9,600 minutes for Line 2 and 19,200 minutes for Line 3. We cannot utilize the production lines over their limitations.

3. Labor Availability

   CALDESCO currently employs 30 craftsmen and assume that an average of 80% of its craftsmen will be available throughout September. Therefore, the total labor availability is going to be 230,400 worker-minutes, which should not be exceeded.

4. Materials Requirement

   The desk production needs materials such as aluminum, particle board and pine sheets. There are resources limitations on these materials, and we should meet these constraints.

5. Production Quotas Requirements

According to CALDESCO's production quotas, there is a range of numbers of different types of desks should be produced. The amount of each desk produced should not go beyond or below the range.

Assumption

We have made the following assumption in this study:

1) CALDESCO will meet all outstanding orders for September, and that it has been able to sell all the desks it produces and maintain its profit margins in part by adhering to a set of in-house quotas.
2) CALDESCO is not concerned with restricting variables to be integers.
3) In sensitivity analysis, if a variable or a constraint varies, we keep all the other parameters unchanged.
4) All the units related to currency are US dollar.

Formulation

The following linear-program formulation was set up based on the objective function and constraints described above. The detailed explanation of the variables and constraints could also be found in the Appendix, with the AMPL solution attached.

<u>Parameters:</u>

Set Pi as the profit for each kind of product {i in 1..9}

- 1: Economy Student Desk
- 2: Economy Standard Desk
- 3: Economy Executive Desk
- 4: Basic Student Desk

- 5: Basic Standard Desk
- 6: Basic Executive Desk
- 7: Hand-crafted Student Desk
- 8: Hand-crafted Standard Desk
- 9: Hand-crafted Executive Desk

Set Oi as the orders of each kind of product {i in 1..9}

Set ALi as aluminum needed for each kind of product {i in 1..9}

Set PBi as particle board needed for each kind of product{i in 1..9}

Set PSi as pine sheets needed for each kind of product {i in 1..9}

Set PL1i as time needed in Production Line 1 for each kind of product {i in 1..9}

Set PL2i as time needed in Production Line 2 for each kind of product {i in 1..9}

Set PL3i as time needed in Production Line 3 for each kind of product {i in 1..9}

Set AFi as time needed for Assembly/Finishing for each kind of product {i in 1..9}

Set Hi as time needed for hand crafting for each kind of product {i in 1..9}

<u>Decision Variables:</u>

Set Xi as the number of each kind of desk to produce {i in 1..9} >= 0

<u>Objective function:</u>

Based on the "company policy/quotas" part in the case, we know that the company has been able to sell all the desks it produces and to maintain its profit margins in part by adhering to a set of in-house quotas. Therefore, we assume CALDESCO will meet all outstanding orders for September as long as it meets all the constraints, and we are going to set the objective as maximizing the total profit, which is the production amount times the profit margin:

Maximize TotalProfit: ΣPi * Xi (i in 1..9) which is 20*X1+30*X2+40*X3+50*X4+80*X5+125*X6+100*X7+250*X8+325*X9

Constraints:

- Meet the minimum order requirement of each kind of product:

  $X_i >= O_i$ (i in 1..9) which is X1 >= 750, X2 >= 1500, X3 >= 100, X4 >= 400, X5 >= 1500, X6 >= 100, X7 >= 25, X8 >= 150, X9 >= 50

- The use of aluminum could not exceed the available amount:

  ΣXi * ALi <= 65,000 (i in 1..9) which is 14*X1+24*X2+30*X3 <= 65000

- The use of particle board could not exceed the available amount:

  ΣXi * PBi <= 60,000 (i in 1..9) which is 8*X1+15*X2+24*X3 <= 60,000

- The use of pine sheets could not exceed the available amount:

  ΣXi * PSi <= 175,000 (i in 1..9) which is

  22*X4+40*X5+55*X6+25*X7+45*X8+60*X9 <= 175000

- Total time for Production Line 1 could not exceed the available amount

  ΣXi * PL1i <= 9,600 (i in 1..9) which is 1.5*X1+2*X2+2.5*X3 <= 9,600

- Total time for Production Line 2 could not exceed the available amount

  ΣXi * PL2i <= 9,600 (i in 1..9) which is X1+X2+X3+X4+X5+X6 <= 9,600

- Total time for Production Line 3 could not exceed the available amount

  ΣXi * PL3i <= 19,200 (i in 1..9) which is 3X4+4*X5+5*X6+3*X7+4*X8+5*X9 <= 19,200

- Total time for labor could not exceed the available amount, where total amount of time required to produce a desk = 2*(total production line time) + (hand crafting time) + (assembly/finishing time):

  ΣXi * (2*PL1i + 2*PL2i + 2*PL3i + AFi + Hi) <=230,400 (i in 1..9) which is 15*X1+17*X2+19*X3+23*X4+28*X5+32*X6+76*X7+93*X8+110*X9 <= 230,400

- The production quantity of Economy desks could not go below its minimum quota limits

  0.2* (ΣXi) <= X1 + X2 + X3 (i in 1..9)

- The production quantity of Economy desks could not exceed its maximum quota limits

  $X1 + X2 + X3 <= 0.5 * (\Sigma Xi)$ (i in 1..9)

- The production quantity of Basic desks could not go below its minimum quota limits

  $0.4 * (\Sigma Xi) <= X4 + X5 + X6$ (i in 1..9)

- The production quantity of Basic desks could not exceed its maximum quota limits

  $X4 + X5 + X6 <= 0.6 * (\Sigma Xi)$ (i in 1..9)

- The production quantity of Hand-crafted desks could not go below its minimum quota limits

  $0.1 * (\Sigma Xi) <= X7 + X8 + X9$ (i in 1..9)

- The production quantity of hand-crafted desks could not exceed its maximum quota limits

  $X7 + X8 + X9 <= 0.2 * (\Sigma Xi)$ (i in 1..9)

- The production quantity of Student desks could not go below its minimum quota limits

  $0.2 * (\Sigma Xi) <= X1+X4+X7$ (i in 1..9)

- The production quantity of Student desks could not exceed its maximum quota limits

  $X1 + X4 + X7 <= 0.35 * (\Sigma Xi)$ (i in 1..9)

- The production quantity of Standard desks could not go below its minimum quota limits

  $0.4 * (\Sigma Xi) <= X2 + X5 + X8$ (i in 1..9)

- The production quantity of Standard desks could not exceed its maximum quota limits

  $X2 + X5 + X8 <= 0.7 * (\Sigma Xi)$ (i in 1..9)

- The production quantity of Executive desks could not go below its minimum quota limits

  $0.05 * (\Sigma Xi) <= X3 + X6 + X9$ (i in 1..9)

- The production quantity of Executive desks could not exceed its maximum quota limit

  $X3 + X6 + X9 <= 0.15 * (\Sigma Xi)$ (i in 1..9)

Solution

The optimal solution of this linear problem is calculated in AMPL based on the objective and constraints listed in the previous section. The following table displays the number of each type of desks Calgary Desk Company should produce in September:

	Student	Standard	Executive
Economy	X1=750	X2=1500	X3=100
Basic	X4=525.537	X5=1657.5	X6=825.403
Hand-Crafted	X7=25	X8=1069.24	X9=50
Optimal Total Profit	612,113		

The following table shows the number of each material, production line and labor usage in September:

	x1	x2	x3	x4	x5	x6	x7	x8	x9	Total	Limit
Aluminum	10500	36000	3000							49500	65000
Particle Board	6000	39780	2400							48180	60000
Pine Sheet				11561.81	66300	45397.16	625	48115.8	3000	17499.79	17500
Product Line 1	1125	3000	250							4375	9600
Production Line 2	750	1500	100	525.537	1657.5	825.403				5358.44	9600
Production Line 3				1576.61	6630	4127.02	75	4276.96	250	16935.59	19200
Labor	11250	25500	1900	12081.35	46410	26412.9	1900	99439.32	5500	230399.57	230400

As we can see from the table, all materials required fall into the limit range.

Discussion

Besides the factors mentioned in the case description, there are several other real-life conditions that we should consider when determining the option solutions in order to maximize the profit. The following parts discuss these real-life factors that CALDESCO should pay attention to in the future. Also If we could include these factors into our model, the result could be more accurate:

1. Inventory

CALDESCO should have its own inventory to store raw materials or products it sells. For example, in the optimal solution, the number of basic student desks we should produce is much higher that the order number. If in some months, CALDESCO is unable to sell all desks it produced, it needs to put the unsalable desks to inventory. Then, we need to consider inventory storage costs and labor costs associated with inventory management into the linear program, which could help us to make better decisions.

2. Workers' availability

CALDESCO currently employs 30 craftsmen and expects that only 80% workforce will be available throughout September, which is relatively low in the real world. Also, in the result of this linear program, the slackness of labor time is close to zero, which means labor is taken full advantage in our program. If CALDESCO could provide overtime pay policy, some craftsmen may be willing to work overtime for extra money. If we could increase work time upper bound, the profit could increase as well. Also in real life, the company should also consider the workforce management for its employees. For example, we don't need all craftmen for everyday work and craftsmen have their shifts for working during the week. Therefore, CALDESCO may need to allocate right number of craftsmen to complete the work.

Determining the right size of resources, here the number of craftsmen is very important to cut the labor cost and improve the efficiency.

3. Quota/policy

In this case, CALDESCO uses maximum and minimum quotas for desk production to meet the goal each month. We assume this in-house quotas was made by CALDESCO's history sale performance. The quota range for each type of desk gives company a goal to manufacturing with the minimum of cost and maximize the profit. However, certain outliers in sale may happen for some unexpected events. For example when the school starts, the demand for student desks will increase and the quota range would not give a proper approximation for production. Therefore, we need to know how CALDESCO derives this quota range and how flexibility the range will be to meet the demand for unexpected events.

Sensitivity Analysis

We want to make the sensitivity analysis by changing some assumptions about the problem that could make sense in practice.

1. Craftsmen's work availability

Constraint	Y	Slack	Lower bound	Upper bound
labor	2.59735	0	200443	239325

Since the slackness of labor time is almost zero and the slackness of the three production lines are high, there are room for more labor and thus, maybe more profit. Also, it is not practical to have such a precise assumption on labor availability. Therefore, we want to analyze how the result would change if we improve the craftsman availability assumption, for example, change it from 80% to 90%. When the the percentage is 90%, the total available time is going to be 259200 minutes. Since 259200 is above the upper bound, we need to rerun the program:

The labor limit constraint would be updated as follows in the new linear program:

subject to Labor_limit:

ΣXi * (2 * PL1[i] + 2 * PL2[i] + 2 * PL3[i] + AF[i] + H[i]) <= 259200 (i in 1..9) which is

15*X1+17*X2+19*X3+23*X4+28*X5+32*X6+76*X7+93*X8+110*X9 <= 259200

And the new result would be:

	Student	Standard	Executive
Economy	863.967	1759.05	100
Basic	472.544	2150.48	100
Hand-Crafted	25	515.378	821.133
Optimal Total Profit	680,429		

The total profit changes from $612,113 to $680,429, by 11% or $68,316. As a result, it is reasonable if the company pays an incentive policy, such as overtime bonus, which is less than $68,316 to increase the total working time of the craftsman and as a result, increase the total profit.

2. Quota

Constraint	Y	Slack	Lower bound	Upper bound
quota student LB	12.7924	01.14E-13	-281.915	114.494

Since the slackness of the lower bound of the student production is almost zero, producing less products under student category and leaving room for other products seems to be appropriate. It is very likely that we can gain more profit by cutting down the quota of production under student category. However, we are still unsure due to the existence of other constraints. To test the sensitivity of the student production quota, we move the lower bound from 20% to 10%.

The student desk lower quota limit constraint would be updated as follows in the new linear program:

subject to Stu_quotas_lower:

0.1 * (ΣXi) <= X1+X4+X7 (i in 1..9)

And the new result would be:

	Student	Standard	Executive
Economy	750	1500	100
Basic	400	1721.47	817.12
Hand-Crafted	25	1083.88	50
Optimal Total Profit	613,578		

The total profit changes from $612,113 to $613,578, by 0.2% or $1,465. As a result, it is reasonable if the company produce less products under student category and leave room for other products.

3. Resource availability - particle board and pine sheet

Constraint	Y	Slack	Lower bound	Upper bound
particle board	0	29100	30900	1.00E+20
pine sheet	0.234495	-2.91E-11	169646	195940

In response to government's new policy on environmental protection by raising related taxes, the wood supplier company raised the price of both particle boards and pine sheets. This may lead to the decrease availability of particle boards and pine sheets for September. The particle board available is decreased by 10% to be 54,000 square feet, and the pine sheet available is decreased by 10% to be 157,500 square feet.

The particle board and pine sheet availability constraint would be updated as follows in the new linear program:

ΣXi * PBi <= 54,000 (i in 1..9) which is 8*X1+15*X2+24*X3 <= 54,000

ΣXi * PSi <= 157,500 (i in 1..9) which is

22*X4+40*X5+55*X6+25*X7+45*X8+60*X9 <= 157500

Now the pine sheet available falls below two the lower bound (169646), we need to rerun the AMPL model.

And the new result would be:

	Student	Standard	Executive
Economy	787.482	1500	100
Basic	400	1919.7	142.75
Hand-Crafted	25	520.87	666.612
Optimal Total Profit	605536		

We get the new objective value 605536. This revenue drop of 612113-605536=6577 compared with the primal problem is reasonable because of the shortage of resource supply.

Besides the above detailed sensitivity analysis on some factors, we also provide the following sensitivity analysis on each variable and constraints in the following tables. This first one shows the corresponding sensitivity analysis of the decision variables:

Products		Current	Reduced Cost	Lower Bound	Upper Bound
Economy	Student	750	0	750	Infinity
	Standard	1500	0	1500	Infinity
	Executive	100	0	100	Infinity
Basic	Student	525.537	0	400	Infinity

	Standard	1657.5	0	1500	Infinity
	Executive	825.403	0	100	Infinity
Hand-crafted	Student	25	0	25	Infinity
	Standard	1069.24	0	150	Infinity
	Executive	50	0	50	Infinity

Here, reduced cost means the amount by which the cost coefficient of a non-basic variable must be lowered for that variable to become basic. Since this question does not provide any information about cost and revenue, the profit of each product is fixed and the reduce cost is alway zero.

Lower bound is the lowest amount of the product that does not turn the basic variable into a non-basic one and vice versa.

Upper bound is the highest amount of the product that does not turn the basic variable into a non-basic one and vice versa.

The following table shows the corresponding sensitivity analysis of the constraints:

Constraint	Y	Slack	Lower bound	Upper bound
labor	2.59735	0	200443	239325
aluminum	0	15500	49500	1.00E+20
particle board	0	29100	30900	1.00E+20
pine sheet	0.234495	-2.91E-11	169646	195940
product line 1	0	5225	4375	1.00E+20
product line 2	0	4241.56	5358.44	1.00E+20
product line 3	0	2264.39	16935.6	1.00E+20
quota economy LB	0	1049.46	-1049.46	1.00E+20
quota economy UB	0	901.343	-901.343	1.00E+20
quota basic LB	0	407.367	-407.367	1.00E+20
quota basic UB	0	893.17	-893.17	1.00E+20

quota hand-cr LB	0	493.976	-493.976	1.00E+20
quota hand-cr UB	0	156.293	-156.293	1.00E+20
quota student LB	12.7924	-1.14E-13	-281.915	114.494
quota student UB	0	975.403	-975.403	1.00E+20
quota standard LB	0	1625.67	-1625.67	1.00E+20
quota standard UB	0	325.134	-325.134	1.00E+20
quota executive LB	0	650.269	-650.269	1.00E+20
quota executive UB	31.0932	-8.53E-14	-331.807	114.905

Slackness measures the resource that is leftover with the current amount of production. That means, if the slack is not zero, no matter how we increase the supply of that resource, the optimal solution will not change because that only introduce more leftover.

The value of 'Y' is the shadow price. The profit will increase by that amount as the resource/quota increases by one unit. Notice that, the shadow price has non-zero value only when the slackness is zero because otherwise we only introduce more leftover and will not affect the optimal solution. As long as the value of constraint is between the lower and upper bounds, the shadow price will not change. Only when the slackness is zero or almost zero (the optimal solution hits one of the bounds), the sensitivity analysis of that constraint is worth to investigate.

Appendix

1. AMPL Model

```
# parameters
param n;
param P{i in 1..n};
# profit of each kind of product
# 1: Economy Student Desk
# 2: Economy Standard Desk
# 3: Economy Executive Desk
# 4: Basic Student Desk
# 5: Basic Standard Desk
# 6: Basic Executive Desk
# 7: Hand-crafted Student Desk
# 8: Hand-crafted Standard Desk
# 9: Hand-crafted Executive Desk
 param O{i in 1..n};
# orders of each kind of product
 param AL{i in 1..n};
# aluminum needed for each kind of product
 param PB{i in 1..n};
# particle board needed for each kind of product
 param PS{i in 1..n};
# pine sheets needed for each kind of product
 param PL1{i in 1..n};
# time needed in Production Line 1 for each kind of product
 param PL2{i in 1..n};
# time needed in Production Line 2 for each kind of product
 param PL3{i in 1..n};
# time needed in Production Line 3 for each kind of product
 param AF{i in 1..n};
# time needed for Assembly/Finishing for each kind of product
 param H{i in 1..n};
# time needed for hand crafting for each kind of product

# decision variables
 var x{i in 1..n} >= 0;
# x1= number of Economy Student Desk to produce
# x2= number of Economy Standard Desk to produce
# x3= number of Economy Executive Desk to produce
# x4= number of Basic Student Desk to produce
# x5= number of Basic Standard Desk to produce
# x6= number of Basic Executive Desk to produce
# x7= number of Hand-crafted Student Desk to produce
# x8= number of Hand-crafted Standard Desk to produce
# x9= number of Hand-crafted Executive Desk to produce

# objective function
 maximize total_profit: sum {i in 1..n} P[i] * x[i];
# Maximize the total profit, which is the sum of every Profit*Production

# constraints
subject to minimum_order
        {i in 1..n}:
        x[i] >= O[i];
# meet the minimum order for each kind of product in order to maintain its profit margin
subject to Al_available:
        sum {i in 1..n} x[i] * AL[i] <= 65000;
# The use of aluminum could not exceed the available amount.
subject to PB_available:
        sum {i in 1..n} x[i] * PB[i] <= 60000;
# The use of particle board could not exceed the available amount.
subject to PS_available:
        sum {i in 1..n} x[i] * PS[i] <= 175000;
# The use of pine sheets could not exceed the available amount.
```

```
subject to PL1_limit:
	sum {i in 1..n} x[i] * PL1[i] <= 9600;
# The total time for Production Line 1 could not exceed the available amount.
subject to PL2_limit:
	sum {i in 1..n} x[i] * PL2[i] <= 9600;
# The total time for Production Line 2 could not exceed the available amount.
subject to PL3_limit:
	sum {i in 1..n} x[i] * PL3[i] <= 19200;
# The total time for Production Line 3 could not exceed the available amount.
subject to Labor_limit:
	sum {i in 1..n} x[i] * (2 * PL1[i] + 2 * PL2[i] + 2 * PL3[i] + AF[i] + H[i]) <= 230400;
# The total time for labor could not exceed the available amount.
subject to Econ_quotas_lower:
	0.2 * sum {i in 1..n} x[i] <= sum {i in 1..3} x[i];
# The production quantity of Economy desks could not below its minimum quota limits
subject to Econ_quotas_upper:
	sum {i in 1..3} x[i] <= 0.5 * sum {i in 1..n} x[i];
# The production quantity of Economy desks could not exceed its maximum quota limits
subject to Bas_quotas_lower:
	0.4 * sum {i in 1..n} x[i]<= sum {i in 4..6} x[i];
# The production quantity of Basic desks could not below its minimum quota limits
subject to Bas_quotas_upper:
	sum {i in 4..6} x[i] <= 0.6 * sum {i in 1..n} x[i];
# The production quantity of basic desks could not exceed its maximum quota limits
subject to Hand_quotas_lower:
	0.1 * sum {i in 1..n} x[i]<= sum {i in 7..9} x[i];
# The production quantity of Hand-crafted desks could not below its minimum quota limits
subject to Hand_quotas_upper:
	sum {i in 7..9} x[i] <= 0.2 * sum {i in 1..n} x[i];
# The production quantity of hand-crafted desks could not exceed its maximum quota limits
subject to Stu_quotas_lower:
	0.2 * sum {i in 1..n} x[i]<= x[1]+x[4]+x[7];
# The production quantity of Student desks could not below its minimum quota limits
subject to Stu_quotas_upper:
	x[1]+x[4]+x[7] <= 0.35 * sum {i in 1..n} x[i];
#The production quantity of Student desks could not exceed its maximum quota limits
subject to Sta_quotas_lower:
	0.4 * sum {i in 1..n} x[i]<= x[2]+x[5]+x[8];
# The production quantity of Standard desks could not below its minimum quota limits
subject to Sta_quotas_upper:
	x[2]+x[5]+x[8] <= 0.7 * sum {i in 1..n} x[i];
#The production quantity of Standard desks could not exceed its maximum quota limits
subject to Exe_quotas_lower:
	0.05 * sum {i in 1..n} x[i]<= x[3]+x[6]+x[9];
# The production quantity of Executive desks could not below its minimum quota limits
subject to Exe_quotas_upper:
	x[3]+x[6]+x[9] <= 0.15 * sum {i in 1..n} x[i];
# The production quantity of Executive desks could not exceed its maximum quota limit
```

2. Data

```
param n := 9;
param P :=
	1 20
	2 30
	3 40
	4 50
	5 80
	6 125
	7 100
	8 250
	9 325;
param O :=
	1 750
	2 1500
	3 100
```

```
        4 400
        5 1500
        6 100
        7 25
        8 150
        9 50;
param AL :=
        1 14
        2 24
        3 30
        4 0
        5 0
        6 0
        7 0
        8 0
        9 0;
param PB :=
        1 8
        2 15
        3 24
        4 0
        5 0
        6 0
        7 0
        8 0
        9 0;
param PS :=
        1 0
        2 0
        3 0
        4 22
        5 40
        6 55
        7 25
        8 45
        9 60;
param PL1 :=
        1 1.5
        2 2.0
        3 2.5
        4 0
        5 0
        6 0
        7 0
        8 0
        9 0;
param PL2 :=
        1 1
        2 1
        3 1
        4 1
        5 1
        6 1
        7 0
        8 0
        9 0;
param PL3 :=
        1 0
        2 0
        3 0
        4 3
        5 4
        6 5
        7 3
        8 4
```

```
        9 5;
param AF :=
        1 10
        2 11
        3 12
        4 15
        5 18
        6 20
        8 25
        9 30;
 param H :=
        1 0
        2 0
        3 0
        4 0
        5 0
        6 0
        7 50
        8 60
        9 70;
```

3. Sensitivity Analysis

```
ampl: reset;
ampl: option solver cplex;
ampl: option cplex_options 'sensitivity';
ampl: model CalgarySym.mod;
ampl: data CalgarySym.dat;
ampl: solve;
CPLEX 12.7.1.0: sensitivity
CPLEX 12.7.1.0: optimal solution; objective 612113.3832
12 dual simplex iterations (2 in phase I)

suffix up OUT;
suffix down OUT;
suffix current OUT;
ampl: display x,x.lb,x.current,x.ub,x.rc;
:    x      x.lb x.current    x.ub       x.rc      :=
1    750      750     20    Infinity    0
2   1500     1500     30    Infinity    0
3    100      100     40    Infinity    0
4    525.537   400     50    Infinity   -2.66454e-15
5   1657.5    1500     80    Infinity   -4.44089e-15
6    825.403   100    125    Infinity    1.42109e-14
7     25       25    100    Infinity    0
8   1069.24    150    250    Infinity   -3.28626e-14
9     50       50    325    Infinity    0
;
ampl: display _conname,_con,_con.slack,_con.current,_con.down,_con.up;
# $4 = _con.current
:       _conname         _con       _con.slack       $4     _con.down
 :=
1    'minimum_order[1]'    -4.06234       0              0       0
2    'minimum_order[2]'   -12.0494        0              0       0
3    'minimum_order[3]'   -38.3373        0              0       0
4    'minimum_order[4]'     0          125.537           0       0
5    'minimum_order[5]'     0          157.501           0       0
6    'minimum_order[6]'     0          725.403           0       0
7    'minimum_order[7]'   -88.3629        0              0       0
8    'minimum_order[8]'     0          919.244           0       0
9    'minimum_order[9]'    -3.7655        0              0       0
10   Al_available          0          15500          65000    49500
11   PB_available          0          29100          60000    30900
12   PS_available          0.234495       0          175000   169646
```

```
13   PL1_limit             0          5225                 0        0
14   PL2_limit             0          4241.56           9600     5358.44
15   PL3_limit             0          2264.39          19200    16935.6
16   Labor_limit           2.59735       -2.91038e-11   230400   200443
17   Econ_quotas_lower      0          1049.46             0    -1049.46
18   Econ_quotas_upper      0           901.343            0     -901.343
19   Bas_quotas_lower       0           407.367            0     -407.367
20   Bas_quotas_upper       0           893.17             0     -893.17
21   Hand_quotas_lower      0           493.976            0     -493.976
22   Hand_quotas_upper      0           156.293            0     -156.293
23   Stu_quotas_lower      12.7924       -5.68434e-14       0     -281.915
24   Stu_quotas_upper       0           975.403            0     -975.403
25   Sta_quotas_lower       0          1625.67             0    -1625.67
26   Sta_quotas_upper       0           325.134            0     -325.134
27   Exe_quotas_lower       0           650.269            0     -650.269
28   Exe_quotas_upper      31.0932        0                0     -331.807
;
:    _con.up      :=
1        0
2        0
3        0
4        0
5        0
6        0
7        0
8        0
9        0
10    1e+20
11    1e+20
12   195940
13       0
14    1e+20
15    1e+20
16   239325
17    1e+20
18    1e+20
19    1e+20
20    1e+20
21    1e+20
22    1e+20
23     114.494
24    1e+20
25    1e+20
26    1e+20
27    1e+20
28     114.905
;

ampl: display _objname,_obj,_varname,_var;
:    _objname      _obj  _varname     _var      :=
1   total_profit   612113   'x[1]'    750
2        .          .     'x[2]'   1500
3        .          .     'x[3]'    100
4        .          .     'x[4]'    525.537
5        .          .     'x[5]'   1657.5
6        .          .     'x[6]'    825.403
7        .          .     'x[7]'     25
8        .          .     'x[8]'   1069.24
9        .          .     'x[9]'     50
;
```